Лекція 14.

CLIPS.

Правила (подовження 2).

Зіставлення зразків з об'єктами

У всіх наведених вище прикладах зразки зіставлялися з фактами зі списку фактів. Крім цього, зразки можна зіставляти з екземплярами об'єктів - екземплярів, певних користувачем класів мовою COOL. Такі зразки називаються зразками об'єктів. Зразки можуть зіставлятися з об'єктами, специфікація яких визначена до створення зразка і які перебувають у границях видимості поточного модуля. Любою клас, що має об'єкти, що відповідають зразку, не може бути вилучений або змінений, поки не буде вилучений зразок. Навіть якщо правило вилучене за допомогою дій, виконуваних у власній правій частині, клас, пов'язаний зі зразком, не може бути змінений доти, поки права частина правила не закінчить роботу. При створенні або видаленні об'єкта всі зразки, що підходять цьому об'єкту, оновлюються. Однак у випадку зміни слоту об'єкта оновлюються тільки ті зразки, які явно зіставляються по цьому слоті. Таким зразком можна використати логічні залежності для обробки змін деяких слотів.
Зміна неактивних слотів або об'єктів неактивних класів не робить ніякого впливу на правила.

Визначення 9.15. Синтаксис зразків об'єктів

```
<зразок об'єкта>       ::=   (object  <атрибута-обмеження>)
<атрибута-обмеження>   ::=   (is-a  <обмеження>)|
                               (name  <обмеження>)|
                               (slot  <обмеження>)
```

Обмеження is-a (є) використовується для визначення обмежень класу, таких як " чи є цей об'єкт екземпляром заданого класу?".
Обмеження is-a також визначає, чи є об'єкт екземпляром класу, що є спадкоємцем класу, заданого в обмеженні, у випадку якщо це не буде явно заборонено зразком.

Обмеження name використовується для визначення конкретного об'єкта із заданим ім'ям. Ім'я, задане в даному обмеженні, повинне бути значенням типу instance-name, а не значенням типу symbol, як звичайно.
Обмеження для складових полів (такі як $?) не можуть використатися з обмеженнями is-a й name. Ці обмеження застосовуються в роботі зі слотами об'єктів так само, як і при роботі зі слотами шаблонів. Як й у випадку зразків для шаблонів, імена слотів для зразка об'єкта повинні бути значеннями типу symbol.

Приведемо кілька прикладів використання зразків об'єктів.

Приклад 9.22. Використання зразків об'єктів

```
(defrule example-1
     (object (is-a MyObjl | MyObj2) )
     =>)
(defrule example-2
     (object (is-a ?x) )
     (object (is-a ~?x) }
     =>)
(defrule example-3
     (object (width ?x&:(> ?x 20)))
     =>)
(defrule example-4
     (object (width ?x) (height ?x))
     =>)
```

Перше правило задовольняє будь-який об'єкт класу MyObj1 або MyObj2. Друге правило активується будь-якою парою об'єктів, що належить різним класам. Третє правило виконується у випадку, якщо буде знайдений об'єкт активного класу, що містить активний слот width, значення якого більше 20. Останній наведений приклад задовольняється будь-яким об'єктом активного класу, що містить активні слоти width й height, значення яких повинні бути рівні.

Адреса зразка

Деякі дії в правій частині правил, такі як retract й unmake-instance, оперують із фактами або об'єктами, що беруть участь у лівій частині. Для того щоб визначити, який факт або об'єкт буде змінюватися, необхідно привласнити змінної адресу конкретного факту або об'єкта. Присвоювання адрес відбувається в лівій частині правила й отримане значення називається *адресою зразка* (pattern-address).

Визначення 9.16. Синтаксис адреси зразка

```
<адреса-зразка> ::= ?<ім'я-змінної> <- <зразок>
```

Стрілка вліво (<-) - необхідна частина синтаксису. Змінна, пов'язана з адресою факту або об'єкта, може рівнятися з інший змінної або використатися зовнішньою функцією. Змінна, пов'язана з адресою факту або об'єкта, може бути також використана для наступного обмеження полів у зразку умовного вираження. Однак не можна зв'язувати змінну в умовному вираженні not.

Як приклад приведемо просте правило, що видаляє всі факти data.

Приклад 9.23. Правило del-data-facts

```
(defrule  del-data-facts
          ?data-facts  <-   (data  $?)
          =>
          (retract  ?data-facts))
```

6.5. 2. Умовний елемент *test*

Умовний елемент test надає можливість накладення додаткових обмежень на слоти фактів або об'єктів. Елемент test задовольняється, якщо викликана в ньому функція повертає значення НЕ-FALSE. Як й у випадку предикатних обмежень зразка в умовному елементі test, можна використати змінні, уже зв'язані зі своїми значеннями. Усередині елемента test можуть бути виконані різні логічні операції, наприклад порівняння змінних.

Визначення 9.17. Синтаксис умовного елемента test

```
<умовний-елемент-test  >   ::=    (test  <виклик-функції>)
```

Вираження test обчислюється щораз при задоволенні інших умовних елементів. Це означає, що умовний елемент test буде обчислений більше одного разу, якщо оброблюване вираження може бути задоволене більш ніж однією групою даних. Використання умовного елемента test може стати причиною автоматичного додавання правилу деяких умовних виражень. Крім того, CLIPS може автоматично з умовні елементи test .

Наведене нижче правило знаходить пару фактів data, причому різниця між значеннями перших полів цих фактів повинна бути більше або рівної 3.

Приклад 9.24. Застосування умовного елемента test

```
(defrule  example
          (data ?x)
          (data ?y)
          (test (>= (abs (- ?y ?x) ) 3))
          =>)
```

Умовний елемент test може привести до автоматичного додавання зразків initial-fact або initial-object у ліву частину правила. Тому не забувайте використати команду reset (яка створює initial-fact й initial-object), щоб бути впевненим у коректній роботі умовного елемента test.

6.5. 3. Умовний елемент *or*

Умовний елемент or дозволяє активувати правило кожним з декількох заданих умовних елементів. Якщо який-небудь із умовних елементів, об'єднаних за допомогою or, удоволений, то й все вираження or уважається вдоволеним. У цьому випадку, якщо всі інші умовні елементи, що входять у ліву частину правила (але не вхідні в or), також задоволені, правило буде активовано. Умовний елемент or може поєднувати будь-яка кількість елементів.

Зауваження

Правило буде активовано для кожного вираження в умовному елементі or, що було задоволено. Таким чином, умовний елемент or робить ефект, ідентичний написанню декількох правил зі схожими посилками й наслідками.

Визначення 9.18. Синтаксис умовного елемента or

```
<умовний-елемент-or > ::= (or <умовний-елемент>+)
```

Приклад 9.25. Застосування умовного елемента or

```
(defrule system-fault
     (error-status unknown) (or (temp high)
     (valve broken)
          (pump off))
           =>
          (printout t "The system has a fault." crlf))
```

Дане правило повідомить про поломку системи, якщо в списку фактів буде присутній факт error-status unknown й один з фактів temp high, valve broken або pump off. У випадку якщо будуть присутні два із цих трьох фактів, наприклад temp high й pump off, те повідомлення буде виведено два рази.

Помітьте, що наведений приклад - точний еквівалент наступних трьох окремих правил:

Приклад 9.26. Еквівалент правилу system-fault

```
(defrule system-fault-1
     (error-status unknown)
     (pump off)
     =>
     (printout t "The system has a fault." crlf))
(defrule  system-fault-2
     (error-status  unknown) (valve broken)
     =>
     (printout t "The system has a fault." crlf))
(defrule system-fault-3
     (error-status unknown) (temp high)
     =>
     (printout t "The system has a fault." crlf))
```

6.5. 4. Умовний елемент *and*

Всі умовні елементи в лівій частині правил CLIPS об'єднані неявним умовним елементом and. Це означає, що всі умовні елементи, задані в лівій частині, повинні задовольнитися, для того щоб правило було активовано. За допомогою явного застосування умовного елемента and можна змішувати різні умови and й or і групувати елементи так, як цього вимагає логіка правил. Умова and задовольняється, тільки якщо всі умови усередині явного and задоволені. У випадку, якщо решта умов у лівій частині правила також щирі, правило буде активовано. Елемент and може поєднувати будь-яке число умовних елементів.

Визначення 9.19. Синтаксис умовного елемента and

```
<умовний-елемент-and> ::= (and <умовн-елемент>+)
```

Приклад 9.27. Застосування умовного елемента and

```
(defrule system-flow
     (error-status confirmed)
     (or   (and  (temp high)
       (valve closed))
(and  (temp low)
     (valve open)))
     =>
     (printout t "The system is having a flow problem. "
     crlf))
```

Якщо умовний елемент and містить умовні елементи test або not як перший елемент, то перед ними автоматично додається зразок initial-fact або initial-object. Пам'ятаємо, що ліва частина будь-якого правила містить неявний елемент and, тому наведене в прикладі 9.28 правило буде автоматично перетворений (див. приклад 9.29).

Приклад 9.28. Правило nothing-to-schedule

```
(defrule  nothing-to-schedule
          (not   (schedule   ?))
          =>
          (printout  t   "Nothing to  schedule."  crlf))
```

Приклад 9.29. Перетворене правило nothing-to-schedule

```
(defrule nothing-to-schedule
          (and (initial-fact)
               (not (schedule ?)))
          =>
          (printout t "Nothing to schedule." crlf))
```

6.5. 5. Умовний елемент *not*

Іноді важливіше відсутність інформації, а не її присутність, тобто виникають ситуації, коли необхідно запустити правило, якщо зразок або інший умовний елемент не задовольняється (наприклад, факт не існує). Умовний елемент not надає цю можливість. Елемент not задовольняється, тільки якщо умовний елемент, що він містить, не задовольняється.

Визначення 9.20. Синтаксис умовного елемента not

```
<умовний-елемент-not>   ::=   (not  <умовний-елемент>)
```

Умовний елемент not може заперечувати тільки одне вираження. Кілька умовних елементів потрібно заперечувати за допомогою декількох елементів not. Ретельно стежите за комбінаціями not з or або and; результат не завжди очевидний!

Приклад 9.30. Застосування умовного елемента not

```
(defrule high-flow-rate
         (temp high)
         (valve open)
         (not (error-status confirmed))
         =>
         (printout t "Recommend closing of valve due to
         high temp" crlf))
```

У логічному елементі not можна використати зв'язані змінні, так само як й в інших умовних елементах:

Приклад 9.31. Правило check-value

```
(defrule check-valve
     (check-status  ?valve)
          (not   (valve-broken  ?valve))
     =>
     (printout  t  "Device   "   ?valve   "   is  OK"
     crlf))
```

За допомогою умовного елемента not можна, нарешті, довести до досконалості наше правило Find-2-coeval-Person.

Якщо ви пам'ятаєте, це правило виводить усілякі пари персон однакового віку. Щоб дане правило не виводило еквівалентні за змістом пари імен (наприклад, Bob-Sue й Sue-Bob), перетворимо нашу програму наступним образом:

Приклад 9.32. Поліпшене правило Find-2-coeval-Person

```
(deftemplate person
          (slot name)
          (slot age))
(deftemplate person-pair
          (slot name1)
          (slot name2)
          (slot age))
(deffacts   people
          (person (name Joe) (age 20))
          (person (name Bob) (age 20))
          (person (name Joe) (age 34))
          (person (name Sue) (age 34))
          (person (name Sue) (age 20)))
```

```
(defrule    Find-2-Coeval-Person
     (person (name ?x) (age ?z))
     (person (name ?ys~?x) (age ?z))
          (not (person-pair (namel ?x) (name2 ?y) (age ?
          z)))
          (not (person-pair (namel ?y) (name2 ?x) (age ?
          z)))
     =>
     (printout t "name=" ?x " name=" ?y " age=" ?z crlf)
     (assert (person-pair (namel ?x) (name2 ?y) (age ?z))))
```

Зверніть увагу на зроблені зміни. По-перше, за допомогою конструктора deftemplate був Доданий додатковий шаблон person-pair. У фактах, що відповідають даному шаблону, буде зберігатися інформація про вже знайдені пари ровесників.

Крім того, було сильно змінене й саме правило. У його лівій частині було додано дві умови:

```
(not (person-pair (name1 ?x) (name2 ?y) (age ?z)))
(not (person-pair (name1 ?y) (name2 ?x) (age ?z)))
```

Ці умовні елементи перевіряють наявність фактів типу person-pair й, тим самим відслідковують, чи була вже оброблена дана пара або її перестановка. Якщо ці факти відсутні, то це означає, що обробка ще не була виконана. У цьому випадку правило активується, і виконуються дії, описані в правій частині правила. А саме виводиться на екран повідомлення про знайдену пару ровесників і додається відповідний факт person-pair, що затверджує, що дана пара вже була оброблена.

Для запуску програми виконайте команди reset й run. Програма виведе на екран наступну інформацію:

Приклад 9.33. Результат роботи правила Find-2-Coeval-Person

```
name=Sue name=Bob  age=20
name=Sue name=Joe  age=20
name=Sue name=Joe  age=34
name=Bob name=Joe  age=20
```

Якщо ви уважно подивитеся на отриманий результат і вихідні дані, то виявите, що це саме те, що нам було потрібно. Це список усіляких ровесників без повторень і з виключенням того факту, що всі люди є ровесниками самі собі. Тепер наше правило досягло повної досконалості! Зверніть увагу на той факт, що якщо ви повторно спробуєте виконати команду run, те нічого не побачите. Це відбувається тому, що в списку фактів утримується інформація про всі оброблені пари, що залишилася після першого запуску. Для того щоб повторно запускати даний приклад, виконуйте команду reset перед кожною командою run.

Умовний елемент not, так само як й test, може привести до автоматичного додавання зразків initial-fact або initial-object у лівій частині правил. Тому не забувайте використати команду reset (яка створює initial-fact й initial-object), щоб бути впевненим у коректній роботі умовного елемента not.

В умовний елемент not, що містить елемент test, автоматично перетвориться в елемент not, що містить and з initial-fact і вихідним елементом test. Наприклад, що випливає умовний елемент із приклада 9.34 перетвориться в елемент із приклада 9.35.

Приклад 9.34. Умовний елемент not, що містить елемент test

```
(not   (test   (>  ?time-l   ?time-2)))
```

Приклад 9.35. Перетворений умовний елемент not, L утримуючий елемент test

```
(not (and   (initial-fact)
               (test   (>  ?time-l  ?time-2))))
```

Зауваження

Помітьте, що найбільш простим і правильним способом запису даного виразу буде:

```
(test   (not   (>  ?time-1  ?time-2) ) ).
```

9.5. 9. Умовний елемент *exists*

Умовний елемент exists дозволяє визначити, чи існує хоча б один набір даних (фактів або об'єктів), які задовольняють умовним елементам, заданим усередині елемента exists.

Визначення 9.21. Синтаксис умовного елемента exists

```
<умовний-елемент-exists>   ::=   (exists <умовн-елемент>+)
```

CLIPS автоматично заміняє exists двома послідовними умовними елементами not. Наприклад, що випливає правило (приклад 9.36) буде перетворено в правило із приклада 9.37.

Приклад 9.36. Правило example

```
(defrule example
         (exists   (a  ?x)    (b  ?x))=>)
```

Приклад 9.37. Перетворене правило example

```
(defrule example
         (not (not (and (a ?x) (b ?x))))=>)
```

Тому що внутрішній спосіб реалізації exists використовує умовний елемент not, то для exists справедливі всі зауваження й обмеження, наведені в попередніх лекціях.

Розглянемо наступний приклад:

Приклад 9.38. Використання умовного елемента exists

```
(deftemplate      hero
          (multislot name)
          (slot status  (default unoccupied)))
(deffacts              goal-and-heroes
          (goal save-the-world)
          (hero   (name Death Defying Man))
          (hero   (name  Stupendous Man))
          (hero   (name  Incredible Man)))
(defrule                save-the-world
          (goal  save-the-world)
          (exists   (hero   (status unoccupied)))
          =>
          (printout t  "The day is  saved."  crlf))
```

Дана програма визначає шаблон - героя, що має складене поле з ім'ям героя й простої поле, що містить статус "не зайнятий" за замовчуванням. Конструктор deffacts визначає трьох нічим не зайнятих героїв і поточну мету - порятунок миру. Правило перевіряє, є в цей момент ця мета, і у випадку позитивної відповіді перевіряє, якщо чи який-небудь ще не зайнятий герой. Якщо всі умовні елементи правила задоволені, воно повідомляє, що мир урятований. Зверніть увагу: незважаючи на те, що в нас всі три герої не зайняті, правило буде активовано тільки один раз.

Тому що спосіб реалізації exists використовує умовний елемент not, те умовний елемент exists може привести до автоматичного додавання зразків initial-fact або initial-object у ліву частину правила. Тому не забувайте використати команду reset (яка створює initial-fact й initial-object), щоб бути впевненим у коректній роботі умовного елемента exists.

6.5. 7. Умовний елемент *forall*

Умовний елемент forall дозволяє визначити, що деяка задана умова виконується для всіх заданих умовних елементів.

Визначення 9.22. Синтаксис умовного елемента forall

```
<умовний-елемент fora11> ::= (forall   <умовний-елемент>
                              <умовний-елемент>+)
```

CLIPS автоматично заміняє f про rail комбінацію умовних елементів not й and. Наприклад, що випливає правило (приклад 9.39) буде перетворено так, як показано в прикладі 9.40.

Приклад 9.39. Правило example

```
(defrule  example
          (forall   (a  ?x)    (b ?x)    (c  ?x))=>)
```

Приклад 9.40. Перетворене правило example

```
(defrule example
          (not      (and   (a ?x)
                    (not   (and    (b ?x)
                                   (c  ?x)))))=>)
```

Розглянемо наступний приклад. Правило all-students-passed визначає, чи пройшли всі студенти читання, правопис й арифметику, використовуючи умову forall:

Приклад 9.41. Правило all-students-passed

```
(defrule all-students-passed
         (forall (student ?name)
                 (reading ?name)
                 (writing ?name)
                 (arithmetic ?name))
         =>
         (printout t "All students passed." crlf))
```

Помітьте, що дане правило задовольняється, поки немає жодного студента. При додаванні факту (student Bob) правило перестає задовольнятися, тому що немає фактів, що підтверджують, що Bob пройшов всі необхідні предмети. Правило не почне задовольнятися й після додавання фактів (reading Bob) і (writing Bob). А от після додавання факту (arithmetic Bob) правило буде активоване й зможе вивести на екран відповідний запис. Якщо додати факт (student John), правило знову перестане задовольнятися, тому що один зі студентів (John) не пройшов всі необхідні предмети. Використовуючи умовний елемент exists, ви без праці зможете змінити це правило так, щоб воно не виконувалося у випадку відсутності студентів.

Тому що реалізація forall використовує умовний елемент not, те forall, так само як й not, test й exists, може привести до автоматичного додавання зразків initial-fact або initial-object у ліву частину правила. **Не** забувайте використати команду reset для коректної роботи цього умовного елемента.

6.5. 8. Умовний елемент *logical*

Умовний елемент logical надає механізм підтримки вірогідності для створених правилом даних (фактів або об'єктів), що задовольняють зразкам. Дані, створені в правій частині правила, можуть мати логічну залежність від даних, що задовольнила зразки в лівій частині правила. Така залежність називається *логічною підтримкою*. Дані можуть залежати від групи даних або декількох груп даних, що задовольнили одне або кілька правил. Якщо віддаляються дані, які підтримують деякі інші дані, то залежні дані також автоматично віддаляються.

Якщо деякі дані створені без логічної підтримки (наприклад, за допомогою конструкторів deffacts, definstance або команди assert, уведеної користувачем або викликаної в правій частині правила), то вважається, що вони мають безумовну підтримку. Безумовна підтримка видаляє всі присутні в цей момент умовні підтримки цих даних (але не видаляє самі дані). Подальша логічна підтримка для даних з безумовною підтримкою ігнорується. Видалення правила, що викликало логічну підтримку для даних, видаляє логічну підтримку, згенеровану цим правилом (але не видаляє дані, якщо в них ще є логічна підтримка, згенерована іншим правилом).

Визначення 9.23. Синтаксис умовного елемента logical

```
<умовний-елемент-1ogica1> ::= (logical
                                    <умовний-елемент>+)
```

Умовний елемент logical групує зразки, так само як це робить and. Дана властивість можна використати при об'єднанні елементів and, or й not. Однак тільки перші *n* зразків правила можуть використатися в умовному елементі logical. Наприклад, що випливає правило записане вірно:

Приклад 9.42. Правильний варіант використання умовного елемента logical

```
(defrule  ok
          (logical (a))
          (logical (b))
          (c)
          =>
          (assert (d)))
```

А таке оголошення правил неприпустиме:

Приклад 9.43. Неправильні варіанти використання умовного елемента logical

```
(defrule not-ok-1
        (logical (a))
        (b)
        (logical (c))
        =>
        (assert (d)))
(defrule not-ok-2
        (a)
        (logical (b))
        (logical (c))
        =>
        (assert (d)))
```

```
(defrule not-ok-3
        (or (a)
        (logical (b)))
        (logical (c))
        =>
        (assert (d)))
```

Розглянемо наступний приклад. Включите перегляд списку фактів за допомогою пункту **Facts Window** меню **Windows,** це допоможе стежити за тим, що відбувається в момент виконання програми.

Приклад 9.44. Використання умовного елемента logical

```
(clear)
(reset)
(defrule example
          (logical   (a))
          (b)
          =>
          (assert (c)))
(assert (a) (b) )
(run)
(retract 2)
(retract 1)
```

По команді run правило example, активоване фактами a й b, додає новий факт із, що має логічну підтримку (залежить) від факту a.

Після видалення факту b за допомогою команди (retract 2) нічого особливого не відбувається, але якщо ми видалимо факт a, те побачимо, що це відразу приведе до видалення пов'язаного з ним факту c.

Як згадувалося , умовний елемент logical може бути використаний для створення даних, які будуть логічно пов'язані зі змінами деяких окремих слотів об'єкта, а не від усього об'єкта цілком. Дану можливість можна використати тільки при роботі з об'єктом. При роботі із шаблонами фактів дану можливість використати не можна, тому що зміни слоту факту фактично приводить до видалення старого факту й додавання нового зі зміненими слотами й індексом.

На відміну від фактів зміни слотів об'єкта виконуються без видалення об'єкта. Це поводження ілюструється наведеним нижче приміром:

Приклад 9.45. Використання умовного елемента logical з об'єктами

```
(clear)
(defclass A (is-a USER)
          (role concrete)
          (pattern-match reactive)
          (slot foo (create-accessor write))
          (slot bar (create-accessor write)))
(defrule  match-A
          (logical (object (is-a A) (foo ?)))
          =>
          (assert (new-fact)))
(make-instance a of A)
(run)
(send [a] put-foo 100)
```

Після виконання команди run правило match-A додає факт new-fact, логічно пов'язаний з конкретним значенням слоту foo об'єкта a. При зміні значення даного слоту факт new-fact автоматично віддаляється зі списку фактів.